Pensieve header: A unified verification notebook for the PPSA project.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
SetOptions[EvaluationNotebook[],
  NotebookEventActions → {
    {"MenuCommand", "Save"} :→ (
      SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\PPSA"];
      Get["MakeVSnips.m"]
     ),
    PassEventsDown → True
   }];
```

```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Initialization

The “degree carrier / filtration parameter” is ℏ, and all “coupling constants” are proportional to it.

TD

```
$TℏD = 5; ℏ /: ℏ^d_. /; d > $TℏD := 0;
$TϵD = 2; ϵ /: ϵ^d_. /; d > $TϵD := 0;
SetAttributes[SS, HoldAll];
SS[ℰ_] := Block[{ℏ, ϵ}, (* Shielded Series *)
  Collect[Normal@Series[ℰ /. T → ⅇ^(ℏ t/2), {ℏ, 0, $TℏD}], ℏ, Together]]
Simp[ℰ_] := Collect[ℰ, _CU | _QU, Expand];
```

DeclareAlgebra

QLImplementation

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x; 0 ** _ = _ ** 0 = 0;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, cp, S, pow,
    gs = Generators /. {opts}, cs = Centrals /. {opts}
  },
  gp = Alternatives @@ gs;
  gp = gp | gp_; (* generators pattern *)
  sr = Thread[gs → Range@Length@gs]; (* sorting rule *)
  cp = Alternatives @@ cs; (* centrals pattern *)
  S[ℰ_] := Collect[ℰ, _U, Expand];
  U_i_[ℰ_] := ℰ /. {t : cp ⧴ t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
  B[U@(x_)_i_, U@(y_)_i_] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
  B[U@(x_)_i_, U@(y_)_j_] /; i =!= j := 0;
  B[U@y_, U@x_] := S[-B[U@x, U@y]];
  x_ ** U[] := x; U[] ** x_ := x;
  (a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, S[a b (x ** y)]];
  (a_ * x_U) ** y_ := S[a (x ** y)]; x_ ** (a_ * y_U) := S[a (x ** y)];
  U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
    U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
  U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := S[c U@Table[l, {n}] ** U@{r}];
  U@{c_. * l : gp, r___} := S[c U[l] ** U@{r}];
  U@{c_, r___} /; FreeQ[c, gp] := S[c U@{r}];
  U@{} = U[];
  U@{l_Plus, r___} := S[U@{#, r} & /@ l];
  U@{l_, r___} := U@{Expand[l], r};
  U[ℰ_NonCommutativeMultiply] := U /@ ℰ;
  𝕆_U[poly_, specs___] := Module[{sp, null, vs, us},
    sp = Replace[{specs}, l_List ⧴ (l → null), {1}];
    vs = Join @@ (First /@ sp);
    us = Join @@ (sp /. (l_ → s_) ⧴ (l /. x_i_ ⧴ x_s));
    S[Total[
        CoefficientRules[poly, vs] /. (p_ → c_) ⧴ c U@(us^p)
      ]] /. x_null ⧴ x
    ];
  pow[ℰ_, 0] = U[]; pow[ℰ_, n_] := pow[ℰ, n - 1] ** ℰ;
  𝕊_U[ℰ_, ss___Rule] := S@Total[
     CoefficientRules[ℰ, First /@ {ss}] /.
      (p_ → c_) ⧴ c NonCommutativeMultiply @@ MapThread[pow, {Last /@ {ss}, p}]]
  ]
```

DeclareMorphism

QLImplementation

```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
  Replace[ongs, (g_ → img_) ⧴ (m[U[g]] = img), {1}];
  m[U[]] = V[];
  m[U[g_i_]] := V_i[m[U@g]];
  m[U[vs__]] := NonCommutativeMultiply @@ (m /@ U /@ {vs});
  m[ℰ_] := Simp[ℰ /. oncs /. u_U ⧴ m[u]]; )
```

Implementing $\mathrm{sl}_2^{\gamma\epsilon}$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[CU@a, CU@y] = -γ CU@y; B[CU@x, CU@a] = -γ CU@x;
B[CU@x, CU@y] = 2 ϵ CU@a - t CU[];
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp // HL,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp // HL
  }] // Timing
```

```
{1.75, {(28 t^2 γ^4 + 116 t γ^5 ϵ + 120 γ^6 ϵ^2) CU[y, y, y, x, x] +
   <<23>> + CU[y, y, y, y, y, a, a, a, a, x, x, x, x], 0}}
```

Implementing $\mathcal{U}_{\gamma\epsilon;\hbar}$

With $q = e^{\hbar\gamma\epsilon}$, $A = e^{-\hbar\epsilon a}$, $T = e^{\hbar t/2}$, and $[f, g]_q := fg - qgf$, our algebra is $\mathcal{U}_{\gamma\epsilon;\hbar} = \langle t, y, a, x\rangle / \mathcal{R}$, where $\mathcal{R} = \left([t, *] = 0,\ [a, y] = -\gamma y,\ [x, y]_q = \left(1 - T^2 A^2\right) / \hbar,\ [x, a] = -\gamma x\right)$.

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q = SS[e^(γ ϵ ℏ)]; (*T=SS[e^(ℏ t/2)];*)
B[QU@a, QU@y] = -γ QU@y; B[QU@x, QU@a] = -γ QU@x;
B[QU@x, QU@y] = (q - 1) QU@{y, x} + 𝕆_QU[SS[(1 - T^2 e^(-2 ϵ a ℏ)) / ℏ], {a}];
```

```
With[{bas = QU /@ {y, a, x}}, Table[{z1, z2} → Simp[z1 ** z2 - z2 ** z1], {z1, bas}, {z2, bas}] ]
```

$$\{\{\{QU[y], QU[y]\} \to 0,\ \{QU[y], QU[a]\} \to \gamma\, QU[y],$$
$$\{QU[y], QU[x]\} \to \left(t + \frac{t^2 \hbar}{2} + \frac{t^3 \hbar^2}{6} + \frac{t^4 \hbar^3}{24} + \frac{t^5 \hbar^4}{120} + \frac{t^6 \hbar^5}{720}\right) QU[\,] +$$
$$\left(-2 \epsilon - 2 t \epsilon \hbar - t^2 \epsilon \hbar^2 - \frac{1}{3} t^3 \epsilon \hbar^3 - \frac{1}{12} t^4 \epsilon \hbar^4 - \frac{1}{60} t^5 \epsilon \hbar^5\right) QU[a] +$$
$$\left(2 \epsilon^2 \hbar + 2 t \epsilon^2 \hbar^2 + t^2 \epsilon^2 \hbar^3 + \frac{1}{3} t^3 \epsilon^2 \hbar^4 + \frac{1}{12} t^4 \epsilon^2 \hbar^5\right) QU[a, a] + \left(-\gamma \epsilon \hbar - \frac{1}{2} \gamma^2 \epsilon^2 \hbar^2\right) QU[y, x]\},$$
$$\{\{QU[a], QU[y]\} \to -\gamma\, QU[y],\ \{QU[a], QU[a]\} \to 0,\ \{QU[a], QU[x]\} \to \gamma\, QU[x]\},$$
$$\{\{QU[x], QU[y]\} \to \left(-t - \frac{t^2 \hbar}{2} - \frac{t^3 \hbar^2}{6} - \frac{t^4 \hbar^3}{24} - \frac{t^5 \hbar^4}{120} - \frac{t^6 \hbar^5}{720}\right) QU[\,] +$$
$$\left(2 \epsilon + 2 t \epsilon \hbar + t^2 \epsilon \hbar^2 + \frac{1}{3} t^3 \epsilon \hbar^3 + \frac{1}{12} t^4 \epsilon \hbar^4 + \frac{1}{60} t^5 \epsilon \hbar^5\right) QU[a] +$$
$$\left(-2 \epsilon^2 \hbar - 2 t \epsilon^2 \hbar^2 - t^2 \epsilon^2 \hbar^3 - \frac{1}{3} t^3 \epsilon^2 \hbar^4 - \frac{1}{12} t^4 \epsilon^2 \hbar^5\right) QU[a, a] + \left(\gamma \epsilon \hbar + \frac{1}{2} \gamma^2 \epsilon^2 \hbar^2\right) QU[y, x],$$
$$\{QU[x], QU[a]\} \to -\gamma\, QU[x],\ \{QU[x], QU[x]\} \to 0\}\}$$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[HL[z1 ** (z2 ** z3) - (z1 ** z2) ** z3] // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple (~34 secs @ T\hbar$D=5, T\epsilon$D=2):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   HL[z1 ** (z2 ** z3) - rhs // Simp ]
  }] // Timing
```

$$\left\{111.047, \left\{\left(28 t^2 \gamma^4 + 116 t \gamma^5 \epsilon + 120 \gamma^6 \epsilon^2 + \ll 12 \gg + \frac{7}{10} t^7 \gamma^4 \hbar^5 + \frac{1549}{90} t^6 \gamma^5 \epsilon \hbar^5 + \frac{2215}{12} t^5 \gamma^6 \epsilon^2 \hbar^5\right) QU[y, \ll 3 \gg, x] + \ll 22 \gg, 0\right\}\right\}$$

Verifying that $\mathrm{Lim}_{\hbar \to 0}\, QU = CU$ using a “random” product (~23 secs @ T\hbar$D=5, T\epsilon$D=2):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   Short[lhs = z1 ** (z2 ** z3)],
   Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
   Expand[Limit[rhs /. {QU → CU, T → ⅇ^(ħ t/2)}, ħ → 0] - lhs] // HL
  }] // Timing
```

$$\{44.4531, \{2\,(8 t^2 \gamma^4 + 16 t \gamma^5 \epsilon)\, CU[y, y, y, x, x] + \ll 94 \gg + CU[y, y, y, y, y, a, a, a, a, x, x, x, x], \ll 1 \gg, 0\}\}$$

Implementing θ

theta

```
DeclareMorphism[Cθ, CU → CU, {y → -CU@x, a → -CU@a, x → -CU@y}, {t → -t, T → T^-1}];
DeclareMorphism[Qθ, QU → QU, {y → 𝕆QU[SS[-T^-1 e^(ħ ϵ a) x], {a, x}],
  a → -QU@a, x → 𝕆QU[SS[-T^-1 e^(ħ ϵ a) y], {a, y}]}, {t → -t, T → T^-1}]
```

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[z → Cθ[z] → HL[Cθ[Cθ[z]]], {z, bas}] ]
```

{CU[y] → -CU[x] → CU[y], CU[a] → -CU[a] → CU[a], CU[x] → -CU[y] → CU[x]}

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cθ[z1 ** z2] - Cθ[z1] ** Cθ[z2] // HL, {z1, bas}, {z2, bas}] ]
```

{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → Qθ[z] → HL[Qθ[Qθ[z]]], {z, bas}] ]
```

$$\Big\{\mathrm{QU}[y]\to\left(-1+\frac{t\hbar}{2}-\frac{t^2\hbar^2}{8}+\frac{t^3\hbar^3}{48}-\frac{t^4\hbar^4}{384}+\frac{t^5\hbar^5}{3840}\right)\mathrm{QU}[x]+\left(-\epsilon\hbar+\frac{1}{2}t\epsilon\hbar^2-\frac{1}{8}t^2\epsilon\hbar^3+\frac{1}{48}t^3\epsilon\hbar^4-\frac{1}{384}t^4\epsilon\hbar^5\right)\mathrm{QU}[a,x]+\left(-\frac{1}{2}\epsilon^2\hbar^2+\frac{1}{4}t\epsilon^2\hbar^3-\frac{1}{16}t^2\epsilon^2\hbar^4+\frac{1}{96}t^3\epsilon^2\hbar^5\right)\mathrm{QU}[a,a,x]\to\mathrm{QU}[y],\ \mathrm{QU}[a]\to-\mathrm{QU}[a]\to\mathrm{QU}[a],$$

$$\mathrm{QU}[x]\to\left(-1+\frac{t\hbar}{2}+\gamma\epsilon\hbar-\frac{t^2\hbar^2}{8}-\frac{1}{2}t\gamma\epsilon\hbar^2-\frac{1}{2}\gamma^2\epsilon^2\hbar^2+\frac{t^3\hbar^3}{48}+\frac{1}{8}t^2\gamma\epsilon\hbar^3+\frac{1}{4}t\gamma^2\epsilon^2\hbar^3-\frac{t^4\hbar^4}{384}-\frac{1}{48}t^3\gamma\epsilon\hbar^4-\frac{1}{16}t^2\gamma^2\epsilon^2\hbar^4+\frac{t^5\hbar^5}{3840}+\frac{1}{384}t^4\gamma\epsilon\hbar^5+\frac{1}{96}t^3\gamma^2\epsilon^2\hbar^5\right)\mathrm{QU}[y]+\left(-\epsilon\hbar+\frac{1}{2}t\epsilon\hbar^2+\gamma\epsilon^2\hbar^2-\frac{1}{8}t^2\epsilon\hbar^3-\frac{1}{2}t\gamma\epsilon^2\hbar^3+\frac{1}{48}t^3\epsilon\hbar^4+\frac{1}{8}t^2\gamma\epsilon^2\hbar^4-\frac{1}{384}t^4\epsilon\hbar^5-\frac{1}{48}t^3\gamma\epsilon^2\hbar^5\right)\mathrm{QU}[y,a]+\left(-\frac{1}{2}\epsilon^2\hbar^2+\frac{1}{4}t\epsilon^2\hbar^3-\frac{1}{16}t^2\epsilon^2\hbar^4+\frac{1}{96}t^3\epsilon^2\hbar^5\right)\mathrm{QU}[y,a,a]\to\mathrm{QU}[x]\Big\}$$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[Qθ[z1 ** z2] - Qθ[z1] ** Qθ[z2]]], {z1, bas}, {z2, bas}] ]
```

{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}

The Asymmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

ADeq

```
AD$f = γ/ℏ e^(ℏ (t/2 - (a+γ) ϵ)) ((Cosh[ℏ (a ϵ + γ ϵ/2 - t/2)] - Cosh[ℏ Sqrt[((t - γ ϵ)/2)^2 + ϵ ω]]) /
    (Sinh[γ ϵ ℏ/2] (a^2 ϵ + a γ ϵ - a t - ω)));
```

```
SS[AD$f] // Simplify
```

$$\frac{1}{1440}\left(1440+720\left(t-2(a+\gamma)\epsilon\right)\hbar+120\left(2t^2-7t(a+\gamma)\epsilon+\epsilon\left(7a^2\epsilon+13a\gamma\epsilon+6\gamma^2\epsilon+\omega\right)\right)\hbar^2+\right.$$
$$60\left(t^3-5t^2(a+\gamma)\epsilon-2(a+\gamma)\epsilon^2\omega+t\epsilon\left(9a^2\epsilon+17a\gamma\epsilon+8\gamma^2\epsilon+\omega\right)\right)\hbar^3+$$
$$2\left(6t^4-39t^3(a+\gamma)\epsilon-t(32a+33\gamma)\epsilon^2\omega+2\epsilon^2\omega^2+t^2\epsilon\left(101a^2\epsilon+192a\gamma\epsilon+91\gamma^2\epsilon+9\omega\right)\right)\hbar^4+$$
$$\left.t\left(2t^4-16t^3(a+\gamma)\epsilon-t(20a+21\gamma)\epsilon^2\omega+2\epsilon^2\omega^2+t^2\epsilon\left(54a^2\epsilon+103a\gamma\epsilon+49\gamma^2\epsilon+4\omega\right)\right)\hbar^5\right)$$

Scaling behaviour of AD$f:

```
Simplify[AD$f == ((AD$f /. γ → 1) /. {ϵ → γ ϵ, a → γ^-1 a, ω → γ^-1 ω})]
```

True

```
FullSimplify[
 AD$f == ((AD$f /. γ → 1) /. {ℏ → γ^2 ℏ, ϵ → ϵ / γ, a → a / γ, t → γ^-2 t, ω → γ^-3 ω})]
```

True

ADeq

```
AD$ω = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a];
```

ADeq

```
DeclareMorphism[AD, QU → CU,
  {a → CU@a, x → CU@x, y → S_CU[SS[AD$f], a → CU[a], ω → AD$ω] ** CU@y}]
```

Verifying that the asymmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[AD[z1 ** z2] - AD[z1] ** AD[z2]]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Symmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

SDeq

$$\mathrm{S}\mathbb{D}\$g = \sqrt{\left(\left(\mathrm{Cosh}\left[\frac{\hbar}{2}\sqrt{t^2+\gamma^2\,\epsilon^2+4\,\epsilon\,\varpi}\right] - \mathrm{Cosh}\left[\frac{\hbar}{2}\left(t-\left(2\,a+\gamma\right)\epsilon\right)\right]\right)\Big/ \left(\mathrm{Sinh}\left[\frac{\gamma\,\epsilon\,\hbar}{2}\right]\left(t\left(2\,a+\gamma\right)-2\,a\,\left(a+\gamma\right)\epsilon+2\,\varpi\right)\hbar\big/\left(2\,\gamma\right)\right)\right)};$$

$$\Big\{\mathrm{S}\mathbb{D}\$P = \frac{\mathrm{Cosh}\left[\hbar\left(\frac{\epsilon-t}{2}+\epsilon\,a\right)\right]-\mathrm{Cosh}\left[\hbar\sqrt{\frac{t^2+\epsilon^2}{4}+\epsilon\,\varpi}\right]}{\hbar\,\mathrm{Sinh}\left[\frac{-\epsilon\,\hbar}{2}\right]\left(\varpi-\epsilon\,a^2+\left(t-\epsilon\right)a+t\big/2\right)},$$

$$\mathrm{Simplify}\left[\mathrm{S}\mathbb{D}\$P == \left(\mathrm{S}\mathbb{D}\$P\ /.\ \{a\to -a-1,\ t\to -t\}\right)\right]\ //\ \mathrm{HL},$$

$$\mathrm{PowerExpand@Simplify}\left[\left(\mathrm{S}\mathbb{D}\$P\ /.\ \left\{\hbar\to\gamma^2\,\hbar,\ \epsilon\to\epsilon/\gamma,\ a\to a/\gamma,\ t\to\gamma^{-2}\,t,\ \varpi\to\gamma^{-3}\,\varpi\right\}\right) == \mathrm{S}\mathbb{D}\$g\ \left(\mathrm{S}\mathbb{D}\$g\ /.\ \{a\to -a-\gamma,\ t\to -t\}\right)\right]\ //\ \mathrm{HL},$$

$$\mathrm{S}\mathbb{D}\$Q = \mathrm{Simplify}\left[\mathrm{S}\mathbb{D}\$P\ /.\ \left\{a\to c-1\big/2\right\}\right],$$

$$\mathrm{Simplify}\left[\mathrm{S}\mathbb{D}\$Q == \left(\mathrm{S}\mathbb{D}\$Q\ /.\ \{c\to -c,\ t\to -t\}\right)\right]\ //\ \mathrm{HL},$$

$$\mathrm{Simplify}\left[\mathrm{S}\mathbb{D}\$g == \mathrm{FullSimplify}\left[\sqrt{\mathrm{S}\mathbb{D}\$Q}\ /.\ c\to a+1\big/2\ /.\ \left\{\hbar\to\gamma^2\,\hbar,\ \epsilon\to\epsilon/\gamma,\ a\to a/\gamma,\ t\to\gamma^{-2}\,t,\ \varpi\to\gamma^{-3}\,\varpi\right\}\right]\right]\ //\ \mathrm{HL}$$

$$\Big\}$$

$$\left\{-\left(\left(\left(\mathrm{Cosh}\left[\left(a\,\epsilon+\frac{1}{2}\left(-t+\epsilon\right)\right)\hbar\right]-\mathrm{Cosh}\left[\sqrt{\frac{1}{4}\left(t^2+\epsilon^2\right)+\epsilon\,\varpi}\ \hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(\frac{t}{2}+a\,\left(t-\epsilon\right)-a^2\,\epsilon+\varpi\right)\hbar\right)\right),\ \mathrm{True},\ \mathrm{True},\right.$$

$$\left.-\left(\left(4\left(\mathrm{Cosh}\left[\frac{1}{2}\left(t-2\,c\,\epsilon\right)\hbar\right]-\mathrm{Cosh}\left[\frac{1}{2}\sqrt{t^2+\epsilon^2+4\,\epsilon\,\varpi}\ \hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(4\,c\,t+\epsilon-4\,c^2\,\epsilon+4\,\varpi\right)\hbar\right)\right),\ \mathrm{True},\ \mathrm{True}\right\}$$

$$\mathrm{FullSimplify}\left[\mathrm{S}\mathbb{D}\$g\ /.\ \left\{\hbar\to\gamma^2\,\hbar,\ \epsilon\to\epsilon/\gamma,\ a\to a/\gamma,\ t\to\gamma^{-2}\,t,\ \varpi\to\gamma^{-3}\,\varpi\right\}\right]$$

$$\sqrt{2}\sqrt{\left(\left(\gamma^2\left(-\mathrm{Cosh}\left[\frac{1}{2}\left(t-\left(2\,a+\gamma^2\right)\epsilon\right)\hbar\right]+\mathrm{Cosh}\left[\frac{1}{2}\sqrt{t^2+\gamma^4\,\epsilon^2+4\,\epsilon\,\varpi}\ \hbar\right]\right)\mathrm{Csch}\left[\frac{1}{2}\gamma^2\,\epsilon\,\hbar\right]\right)\Big/\left(\left(t\left(2\,a+\gamma^2\right)-2\,a\,\left(a+\gamma^2\right)\epsilon+2\,\varpi\right)\hbar\right)\right)}$$

$$\mathrm{SS}[\mathrm{S}\mathbb{D}\$g]$$

$$1+\frac{1}{48}\left(t^2-2\,a\,t\,\epsilon-t\,\gamma\,\epsilon+2\,a^2\,\epsilon^2+2\,a\,\gamma\,\epsilon^2+2\,\epsilon\,\varpi\right)\hbar^2+\frac{1}{23\,040}\left(t^4-4\,a\,t^3\,\epsilon-2\,t^3\,\gamma\,\epsilon+16\,a^2\,t^2\,\epsilon^2+16\,a\,t^2\,\gamma\,\epsilon^2-5\,t^2\,\gamma^2\,\epsilon^2+4\,t^2\,\epsilon\,\varpi+8\,a\,t\,\epsilon^2\,\varpi+4\,t\,\gamma\,\epsilon^2\,\varpi+12\,\epsilon^2\,\varpi^2\right)\hbar^4$$

SDeq

$$\mathrm{S}\mathbb{D}\$f = \mathrm{FullSimplify}\left[e^{\hbar\,(t/2-\epsilon\,a)}\left(\mathrm{S}\mathbb{D}\$g\ /.\ \{a\to -a,\ t\to -t\}\right)\right];$$

```
SS[SD$f]
```

$$1+\frac{1}{2}(t-2a\epsilon)\hbar+\frac{1}{48}\left(7t^2-26at\epsilon+t\gamma\epsilon+26a^2\epsilon^2-2a\gamma\epsilon^2+2\epsilon\varpi\right)\hbar^2+\frac{1}{96}(t-2a\epsilon)\left(3t^2-10at\epsilon+t\gamma\epsilon+10a^2\epsilon^2-2a\gamma\epsilon^2+2\epsilon\varpi\right)\hbar^3+\frac{1}{23040}\left(121t^4-844at^3\epsilon+62t^3\gamma\epsilon+2296a^2t^2\epsilon^2-376at^2\gamma\epsilon^2-5t^2\gamma^2\epsilon^2+124t^2\epsilon\varpi-472at\epsilon^2\varpi-4t\gamma\epsilon^2\varpi+12\epsilon^2\varpi^2\right)\hbar^4+\frac{1}{46080}(t-2a\epsilon)\left(33t^4-220at^3\epsilon+22t^3\gamma\epsilon+584a^2t^2\epsilon^2-136at^2\gamma\epsilon^2-5t^2\gamma^2\epsilon^2+44t^2\epsilon\varpi-152at\epsilon^2\varpi-4t\gamma\epsilon^2\varpi+12\epsilon^2\varpi^2\right)\hbar^5$$

SDeq

```
SD$ϖ = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a] - t γ CU[] / 2;
```

SDeq

```
DeclareMorphism[SD, QU → CU, {a → CU@a,
  x → S_CU[SS[SD$f], a → CU[a], ϖ → SD$ϖ] ** CU@x,
  y → S_CU[SS[SD$g] , a → CU[a], ϖ → SD$ϖ] ** CU@y
 }]
```

Verifying the θ-symmetry:

```
Table[HL@Simplify[Cθ[SD[z]] == SD[Qθ[z]]], {z, QU /@ {y, a, x}}]
```

```
{True, True, True}
```

Verifying that the symmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL@Simp[SD[z1 ** z2] - SD[z1] ** SD[z2]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

R in QU.

Quesne's formula:

Quesne

$$\mathbb{e}_{q\_,n\_}[x\_] := \text{Exp}\left[\sum_{k=1}^{n}\frac{(1-q)^k}{k\,(1-q^k)}x^k\right];\quad \mathbb{e}_{q\_}[x\_] := \mathbb{e}_{q,\$T\epsilon D}[x]$$

```
Table[Together@SeriesCoefficient[e_{ρ,5}[x], {x, 0, n}], {n, 0, 5}]
```

$$\left\{1,1,\frac{1}{1+\rho},\frac{1}{(1+\rho)(1+\rho+\rho^2)},\frac{1}{(1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)},1\Big/\left((1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)(1+\rho+\rho^2+\rho^3+\rho^4)\right)\right\}$$

```
Table[HL@FunctionExpand[QFactorial[n, ρ] SeriesCoefficient[𝕖_{ρ,5}[x], {x, 0, n}]], {n, 0, 5}]
{1, 1, 1, 1, 1, 1}
```

```
QU[R_{i_,j_}] := 𝕆_QU[SS[𝕖^{ħ b1 a2} 𝕖_q[ħ y1 x2] /. b1 → γ^-1 (ϵ a1 - t_i)], {y1, a1} → i, {a2, x2} → j]
```

QU[$R_{3,4}$]

$$\text{QU}[\,] + \frac{\epsilon\,\hbar\,\text{QU}[a_3, a_4]}{\gamma} + \hbar\,\text{QU}[y_3, x_4] + \frac{\epsilon^2\,\hbar^2\,\text{QU}[a_3, a_3, a_4, a_4]}{2\,\gamma^2} + \frac{\epsilon\,\hbar^2\,\text{QU}[y_3, a_3, a_4, x_4]}{\gamma} +$$

$$\left(\frac{\hbar^2}{2} - \frac{1}{4}\,\gamma\,\epsilon\,\hbar^3\right)\text{QU}[y_3, y_3, x_4, x_4] + \frac{\epsilon^2\,\hbar^3\,\text{QU}[y_3, a_3, a_3, a_4, a_4, x_4]}{2\,\gamma^2} +$$

$$\left(\frac{\epsilon\,\hbar^3}{2\,\gamma} - \frac{\epsilon^2\,\hbar^4}{4}\right)\text{QU}[y_3, y_3, a_3, a_4, x_4, x_4] + \left(\frac{\hbar^3}{6} - \frac{1}{4}\,\gamma\,\epsilon\,\hbar^4\right)\text{QU}[y_3, y_3, y_3, x_4, x_4, x_4] +$$

$$\frac{\epsilon^2\,\hbar^4\,\text{QU}[y_3, y_3, a_3, a_3, a_4, a_4, x_4, x_4]}{4\,\gamma^2} + \left(\frac{\epsilon\,\hbar^4}{6\,\gamma} - \frac{\epsilon^2\,\hbar^5}{4}\right)\text{QU}[y_3, y_3, y_3, a_3, a_4, x_4, x_4, x_4] +$$

$$\left(\frac{\hbar^4}{24} - \frac{1}{8}\,\gamma\,\epsilon\,\hbar^5\right)\text{QU}[y_3, y_3, y_3, y_3, x_4, x_4, x_4, x_4] + \frac{\epsilon^2\,\hbar^5\,\text{QU}[y_3, y_3, y_3, a_3, a_3, a_4, a_4, x_4, x_4, x_4]}{12\,\gamma^2} +$$

$$\frac{\epsilon\,\hbar^5\,\text{QU}[y_3, y_3, y_3, y_3, a_3, a_4, x_4, x_4, x_4, x_4]}{24\,\gamma} + \frac{1}{120}\,\hbar^5\,\text{QU}[y_3, y_3, y_3, y_3, y_3, x_4, x_4, x_4, x_4, x_4] -$$

$$\frac{\hbar\,\text{QU}[a_4]\,t_3}{\gamma} - \frac{\epsilon\,\hbar^2\,\text{QU}[a_3, a_4, a_4]\,t_3}{\gamma^2} - \frac{\hbar^2\,\text{QU}[y_3, a_4, x_4]\,t_3}{\gamma} - \frac{\epsilon^2\,\hbar^3\,\text{QU}[a_3, a_3, a_4, a_4, a_4]\,t_3}{2\,\gamma^3} -$$

$$\frac{\epsilon\,\hbar^3\,\text{QU}[y_3, a_3, a_4, a_4, x_4]\,t_3}{\gamma^2} - \frac{\epsilon^2\,\hbar^4\,\text{QU}[y_3, a_3, a_3, a_4, a_4, a_4, x_4]\,t_3}{2\,\gamma^3} -$$

$$\frac{\epsilon^2\,\hbar^5\,\text{QU}[y_3, y_3, a_3, a_3, a_4, a_4, a_4, x_4, x_4]\,t_3}{4\,\gamma^3} - \frac{\epsilon\,\hbar^5\,\text{QU}[y_3, y_3, y_3, a_3, a_4, a_4, x_4, x_4, x_4]\,t_3}{6\,\gamma^2} -$$

$$\frac{\hbar^5\,\text{QU}[y_3, y_3, y_3, y_3, a_4, x_4, x_4, x_4, x_4]\,t_3}{24\,\gamma} + \frac{\hbar^2\,\text{QU}[a_4, a_4]\,t_3^2}{2\,\gamma^2} +$$

$$\frac{\epsilon\,\hbar^3\,\text{QU}[a_3, a_4, a_4, a_4]\,t_3^2}{2\,\gamma^3} + \frac{\hbar^3\,\text{QU}[y_3, a_4, a_4, x_4]\,t_3^2}{2\,\gamma^2} + \frac{\epsilon^2\,\hbar^4\,\text{QU}[a_3, a_3, a_4, a_4, a_4, a_4]\,t_3^2}{4\,\gamma^4} +$$

$$\frac{\epsilon\,\hbar^4\,\text{QU}[y_3, a_3, a_4, a_4, a_4, x_4]\,t_3^2}{2\,\gamma^3} + \frac{\epsilon^2\,\hbar^5\,\text{QU}[y_3, a_3, a_3, a_4, a_4, a_4, a_4, x_4]\,t_3^2}{4\,\gamma^4} +$$

$$\frac{\epsilon\,\hbar^5\,\text{QU}[y_3, y_3, a_3, a_4, a_4, a_4, x_4, x_4]\,t_3^2}{4\,\gamma^3} + \frac{\hbar^5\,\text{QU}[y_3, y_3, y_3, a_4, a_4, x_4, x_4, x_4]\,t_3^2}{12\,\gamma^2} -$$

$$\frac{\hbar^3\,\text{QU}[a_4, a_4, a_4]\,t_3^3}{6\,\gamma^3} - \frac{\epsilon\,\hbar^4\,\text{QU}[a_3, a_4, a_4, a_4, a_4]\,t_3^3}{6\,\gamma^4} - \frac{\hbar^4\,\text{QU}[y_3, a_4, a_4, a_4, x_4]\,t_3^3}{6\,\gamma^3} -$$

$$\frac{\epsilon^2\,\hbar^5\,\text{QU}[a_3, a_3, a_4, a_4, a_4, a_4, a_4]\,t_3^3}{12\,\gamma^5} - \frac{\epsilon\,\hbar^5\,\text{QU}[y_3, a_3, a_4, a_4, a_4, a_4, x_4]\,t_3^3}{6\,\gamma^4} -$$

$$\frac{\hbar^5\,\text{QU}[y_3, y_3, a_4, a_4, a_4, x_4, x_4]\,t_3^3}{12\,\gamma^3} + \frac{\hbar^4\,\text{QU}[a_4, a_4, a_4, a_4]\,t_3^4}{24\,\gamma^4} +$$

$$\frac{\epsilon\,\hbar^5\,\text{QU}[a_3, a_4, a_4, a_4, a_4, a_4]\,t_3^4}{24\,\gamma^5} + \frac{\hbar^5\,\text{QU}[y_3, a_4, a_4, a_4, a_4, x_4]\,t_3^4}{24\,\gamma^4} - \frac{\hbar^5\,\text{QU}[a_4, a_4, a_4, a_4, a_4]\,t_3^5}{120\,\gamma^5} +$$

$$\text{QU}[y_3, y_3, a_4, x_4, x_4]\left(-\frac{\hbar^3\,t_3}{2\,\gamma} + \frac{1}{4}\,\epsilon\,\hbar^4\,t_3\right) + \text{QU}[y_3, y_3, y_3, a_4, x_4, x_4, x_4]\left(-\frac{\hbar^4\,t_3}{6\,\gamma} + \frac{1}{4}\,\epsilon\,\hbar^5\,t_3\right) +$$

$$\text{QU}[y_3, y_3, a_3, a_4, a_4, x_4, x_4]\left(-\frac{\epsilon\,\hbar^4\,t_3}{2\,\gamma^2} + \frac{\epsilon^2\,\hbar^5\,t_3}{4\,\gamma}\right) + \text{QU}[y_3, y_3, a_4, a_4, x_4, x_4]\left(\frac{\hbar^4\,t_3^2}{4\,\gamma^2} - \frac{\epsilon\,\hbar^5\,t_3^2}{8\,\gamma}\right)$$

$R_{1,2}$

$R_{1,2}$

```
R_{1,2} ** R_{1,3} ** R_{2,3}
```

$IterationLimit: Iteration limit of 4096 exceeded.

```
Hold[R_{1,2} ** R_{1,3} ** R_{2,3}]
```

```
ff ** gg
```

$IterationLimit: Iteration limit of 4096 exceeded.

```
Hold[ff ** gg]
```

```
QU[R_{1,2} ** R_{1,3} ** R_{2,3}]
```

$IterationLimit: Iteration limit of 4096 exceeded.

```
QU[Hold[R_{1,2} ** R_{1,3} ** R_{2,3}]]
```

$\mathbb{E}$

$\mathbb{E}[\omega, L, Q, P]$ means $\omega e^{L+Q} P$, where ω is a scalar, L is linear in the a's, Q is a combination of $x_i\, y_j$, and P is a perturbation polynomial.